


CAST3M – ÉLÉMENTS DE THÉORIE

S. PASCAL

STATIQUE



ÉDITION 2020

mél @ : serge.pascal@cea.fr

Documentation Cast3M

<http://www-cast3m.cea.fr>

Cast3M est un logiciel de calcul par la méthode des éléments finis pour la mécanique des structures et des fluides. Cast3M est développé au Département de Modélisation des Systèmes et Structures (DM2S) de la Direction des Énergies du Commissariat à l'Énergie Atomique et aux Énergies Alternatives (CEA).

Le développement de Cast3M entre dans le cadre d'une activité de recherche dans le domaine de la mécanique dont le but est de définir un instrument de haut niveau, pouvant servir de support pour la conception, le dimensionnement et l'analyse de structures et de composants.

Dans cette optique, Cast3M intègre non seulement les processus de résolution (solveur) mais également les fonctions de construction du modèle (pré-processeur) et d'exploitation des résultats (post-traitement). Cast3M est un logiciel « boîte à outils » qui permet à l'utilisateur de développer des fonctions répondant à ses propres besoins.

Cast3M est notamment utilisé dans le secteur de l'énergie nucléaire, comme outil de simulation ou comme plateforme de développement d'applications spécialisées. En particulier, Cast3M est utilisé par l'Institut de Radioprotection et de Sécurité Nucléaire (IRSN) dans le cadre des analyses de sûreté des installations nucléaires françaises.



Avant propos

L'objectif de ce document est de rappeler aux utilisateurs de Cast3M quelques éléments de théorie concernant la mécanique des milieux continus et la méthode des éléments finis afin qu'ils aient une meilleure maîtrise des modélisations qu'ils réalisent avec le logiciel. Ce document s'adresse ainsi à des utilisateurs de Cast3M quelque peu expérimentés, qui possèdent de bonnes connaissances en mécanique et en calcul par éléments finis.

Par souci de simplicité, on assume certaines approximations dans la présentation de ces éléments, notamment dans les développements mathématiques associés. Pour une présentation rigoureuse et détaillée des fondements théoriques du calcul par éléments finis en mécanique des structures, on renvoie le lecteur aux nombreux ouvrages disponibles dans ce domaine.

Enfin, ce document ne propose pas un état de l'art exhaustif des concepts et méthodes numériques utilisés dans Cast3M, notamment dans la procédure de calcul PASAPAS. La raison en est simplement que cet état évolue continûment, au gré des améliorations apportées au logiciel, et que la mise à jour de ce document ne fait pas partie du processus de développement de Cast3M.



Table des matières

1	Équilibre mécanique	9
1.1	Équation de la statique	9
1.2	Formulation faible de l'équilibre	9
1.3	Discrétisation par éléments finis	9
1.4	Formulation élément finis de l'équilibre	9
1.5	Résidu	10
1.6	Opérateurs de Cast3M associés	10
2	Comportement, équilibre et déplacements	11
2.1	Loi de comportement	11
2.2	Relation force-déplacement	11
2.3	Remarque sur les efforts intérieurs	11
2.4	Opérateurs de Cast3M associés	12
3	Conditions sur les déplacements	13
3.1	Relations linéaires entre inconnues de déplacement	13
3.2	Traitement des conditions sur les déplacements	13
3.3	Méthode des multiplicateurs de Lagrange	13
3.4	Opérateurs de Cast3M associés	14
4	Résolution de l'équilibre	15
4.1	Méthode	15
4.2	Incréments de déplacement	15
4.3	Formulation incrémentale des conditions sur les déplacements	15
4.4	Formulation incrémentale de l'équilibre	16
4.5	Convergence	16
4.6	Méthode de minimisation du résidu	16
4.7	Opérateurs de Cast3M associés	17
5	Application	19
5.1	Exemple : statique1.dgibi	19
5.1.1	Description	19
5.1.2	Mise en données	20
5.1.3	Exécution	20
A	Jeux de données Cast3M	23
A.1	statique1.dgibi	23



Chapitre 1

Équilibre mécanique

1.1 Équation de la statique

Soit un domaine matériel Ω soumis à une densité volumique d'efforts f , une densité surfacique d'efforts t sur son bord $\partial\Omega_t$ et des déplacements imposés d sur son bord $\partial\Omega_d$, avec $\partial\Omega_t \cap \partial\Omega_d = \emptyset$.

Dans le cadre de la mécanique des milieux continus, on dit que le domaine matériel Ω est en équilibre lorsque le système d'équations suivant est vérifié :

$$\begin{cases} \operatorname{div} \sigma + f = 0 & \text{sur } \Omega \\ \sigma \cdot n = t & \text{sur } \partial\Omega_t \\ u = d & \text{sur } \partial\Omega_d \end{cases}, \quad (1.1)$$

σ et u étant les champs de contrainte et de déplacement.

1.2 Formulation faible de l'équilibre

En multipliant chaque terme de la première équation du système (1.1) par un champ de déplacement virtuel v non nul, puis en intégrant par partie sur Ω , on obtient une formulation faible de l'équilibre :

$$\int_{\partial\Omega_t} t v dS + \int_{\partial\Omega_d} \sigma n v dS - \int_{\Omega} \sigma \nabla v dV + \int_{\Omega} f v dV = 0, \quad (1.2)$$

où l'on a remplacé le produit $\sigma \cdot n$ par la densité surfacique d'efforts t sur $\partial\Omega_t$ et avec ∇v , le gradient de v .

1.3 Discrétisation par éléments finis

On réalise un maillage Ω^h du domaine Ω . Sur ce maillage, le champ de déplacement v est discrétisé sur la base des fonctions d'interpolation $\langle N_i \rangle$ associées aux éléments du maillage :

$$v^h(x) = \sum_i v_i N_i(x), \quad (1.3)$$

v_i ayant la valeur du déplacement v au point i du maillage.

1.4 Formulation élément finis de l'équilibre

En injectant cette discrétisation dans la formulation faible de l'équilibre (1.2), nous obtenons :

$$\int_{\partial\Omega_t^h} t v_i N_i dS + \int_{\partial\Omega_d^h} \sigma \cdot n v_i N_i dS - \int_{\Omega^h} \sigma v_i \nabla N_i dV + \int_{\Omega^h} f v_i N_i dV = 0. \quad (1.4)$$

Ceci devant être vérifié pour tout v , nous vérifions, à chaque nœud i du maillage, l'égalité suivante :

$$\underbrace{\int_{\partial\Omega_i^h} tNdS}_{F^S} + \underbrace{\int_{\partial\Omega_i^h} \sigma.nNdS}_{F^R} + \underbrace{\int_{\Omega^h} fNdV}_{F^V} - \underbrace{\int_{\Omega^h} \sigma\nabla NdV}_{B.\sigma} = 0,$$

où l'on a négligé l'indice i .

Cette dernière équation fait apparaître les *forces nodales équivalentes* :

- F^S : à la densité surfacique d'efforts t ;
- F^R : à la densité surfacique d'efforts de réaction aux déplacements imposés d ;
- F^V : à la densité volumique d'efforts f ;
- $B.\sigma$: à la densité volumique d'efforts intérieurs.

On note que ces forces nodales équivalentes sont des *quantités intégrées sur le maillage*.

La formulation « éléments finis » de l'équilibre s'écrit donc :

$$\underbrace{F^S + F^R}_{F^{ext}} + \underbrace{F^V - B.\sigma}_{F^{int}} = 0, \quad (1.5)$$

F^{ext} représentant les efforts extérieurs appliqués au domaine matériel Ω et F^{int} , les efforts intérieurs.

1.5 Résidu

Les termes de l'équation d'équilibre (1.5) forment le *résidu* R :

$$R = F^S + F^R + F^V - B.\sigma. \quad (1.6)$$

Numériquement, l'équilibre n'est satisfait que de façon approchée. Ainsi, on peut considérer qu'il est atteint lorsqu'une norme de ce résidu devient négligeable devant une valeur de référence des efforts mis en jeu dans le problème considéré :

$$|R| < \zeta F^{ref}.$$

La valeur de ζ est fournie par l'utilisateur : c'est la précision du calcul.

La valeur de F^{ref} peut être une norme des efforts extérieurs F^{ext} .

Dans certains cas : dilatation libre, décharge... ces derniers peuvent être nuls à l'équilibre. L'utilisateur doit alors également préciser la valeur de F^{ref} ou fournir une tolérance sur les efforts : $F^{tol} = \zeta F^{ref}$.

Si le problème fait intervenir des coques, poutres, tuyaux... il peut aussi s'avérer nécessaire de fournir une tolérance sur les moments M^{tol} .

1.6 Opérateurs de Cast3M associés

Les termes de l'équation (1.5) peuvent être obtenus à l'aide des opérateurs suivants de Cast3M :

- F^S : PRESSion, FORCE, FSUR (Force SURfacique)... selon le cas ;
- F^R : REACtions ;
- F^V : CNEQ (Charge Nodale EQuivalente) ;
- $B.\sigma$: BSIGma ;
- F^{tol}, M^{tol} : voir entrées de la procédure PASAPAS.



Chapitre 2

Comportement, équilibre et déplacements

2.1 Loi de comportement

La loi de comportement relie la contrainte σ à la déformation $\mathcal{D}(u)$:

$$\sigma = \mathcal{C}(\mathcal{D}(u), \varepsilon^{in}, \nu, p), \quad (2.1)$$

\mathcal{D} étant une mesure de la déformation du domaine matériel, ε^{in} , des déformations inélastiques, ν , des variables internes, propres à la loi : variables d'écroutissage, d'endommagement... et p , un certain nombre de paramètres externes : température, taux d'irradiation, etc.

Dans l'expression de la loi de comportement, on peut distinguer un terme reliant linéairement la contrainte au déplacement, en associant le comportement élastique linéaire de la structure à une mesure linéaire de sa déformation, communément appelée mesure en *petites déformations* :

$$\sigma = \mathcal{E} : \nabla_s u + \sigma^{nl}, \quad (2.2)$$

\mathcal{E} étant le tenseur d'élasticité, $\nabla_s u$, le gradient symétrisé des déplacements, et σ^{nl} , le terme non-linéaire complémentaire.

2.2 Relation force-déplacement

En injectant l'expression (2.2) dans la formulation éléments finis de l'équilibre (1.4), nous obtenons :

$$\int_{\partial\Omega^h} t v_i N_i dS + \int_{\partial\Omega_d^h} t_i v_i dS - \int_{\Omega^h} u_j \nabla N_j \mathcal{E} \nabla N_i v_i dV - \int_{\Omega^h} \sigma^{nl} v_i \nabla N_i dV + \int_{\Omega^h} f v_i N_i dV = 0. \quad (2.3)$$

Ceci devant être vérifié quel que soit v , nous pouvons simplifier par v_i pour obtenir :

$$\underbrace{\int_{\partial\Omega^h} t N dS}_{F^S} + \underbrace{\int_{\partial\Omega_d^h} t N dS}_{F^R} + \underbrace{\int_{\Omega^h} f N dV}_{F^V} - \underbrace{\int_{\Omega^h} \sigma^{nl} \nabla N dV}_{B \cdot \sigma^{nl}} = \underbrace{\int_{\Omega^h} u_j \nabla N_j \mathcal{E} \nabla N_i dV}_{K^e \cdot U},$$

K^e étant la raideur élastique et U , le déplacement aux noeuds du maillage. On obtient, finalement :

$$K^e \cdot U = F^S + F^R + F^V - B \cdot \sigma^{nl}. \quad (2.4)$$

2.3 Remarque sur les efforts intérieurs

En faisant passer le terme de gauche de cette dernière équation à droite du signe égal, on obtient :

$$F^S + F^R + F^V - (K^e \cdot U + B \cdot \sigma^{nl}) = 0.$$

À l'aide de l'équation d'équilibre (1.5), on identifie alors que :

$$B.\sigma = K^e.U + B.\sigma^{nl}, \quad (2.5)$$

soit l'équation (2.2) à laquelle on a appliqué l'opérateur « BSIGma ».

2.4 Opérateurs de Cast3M associés

Dans Cast3M, les différents termes ci-dessus peuvent être obtenus avec les opérateurs suivants :

- K^e : RIGIdité;
- σ : COMPortement;
- $\nabla_s u$: EPSIlon, les déformations étant généralement notées ε ;
- \mathcal{E} : ε : ELASticité;
- U : RESOudre, en fournissant à l'opérateur la matrice de raideur K^e et le second membre ($F^S + F^R + F^V - B.\sigma^{nl}$) obtenu avec les opérateurs « + » et « - ».



Chapitre 3

Conditions sur les déplacements

3.1 Relations linéaires entre inconnues de déplacement

En pratique, on a besoin de généraliser la condition de déplacement imposé à des relations linéaires entre inconnues de déplacement, par exemple, pour prescrire des conditions de symétrie. Ainsi, les conditions sur les déplacements s'écrivent de façon plus générale :

$$A.U = d, \quad (3.1)$$

avec A , la matrice des relations linaires entre inconnues de déplacement, et d , les valeurs imposées. Dans le cas d'une condition de déplacement imposé usuelle, A est la matrice identité.

3.2 Traitement des conditions sur les déplacements

Il existe principalement trois méthodes pour résoudre un système linéaire avec des conditions imposées sur ses inconnues primales :

- la méthode d'élimination ;
- la méthode de pénalisation ;
- la méthode des multiplicateurs de Lagrange.

Dans Cast3M, on utilise formellement la méthode des multiplicateurs de Lagrange.

3.3 Méthode des multiplicateurs de Lagrange

Pour un système linéaire d'inconnue U à résoudre tel que :

$$K^e.U = F^S + F^V - B.\sigma^{nl},$$

l'ajout de n conditions sur le champ solution équivaut à introduire n inconnues supplémentaires λ , appelées *multiplicateurs de Lagrange*, telles que :

$$K^e.U + A^T.\lambda = F^S + F^V - B.\sigma^{nl}, \quad \text{avec} \quad A.U = d. \quad (3.2)$$

L'équation (2.4) permet d'identifier que le terme $A^T.\lambda$ est égal, au signe près, aux efforts de réaction aux déplacements imposés F^R . La matrice A étant sans dimension, les multiplicateurs de Lagrange sont homogènes à des efforts.

On peut assembler ces deux équations sous forme matricielle :

$$\begin{pmatrix} K^e & A^T \\ A & 0 \end{pmatrix} \begin{pmatrix} U \\ \lambda \end{pmatrix} = \begin{pmatrix} F^S + F^V - B.\sigma^{nl} \\ d \end{pmatrix}, \quad (3.3)$$

pour expliciter formellement un seul système linéaire :

$$\hat{K} \cdot \hat{U} = \hat{F}. \quad (3.4)$$

La méthode des multiplicateurs de Lagrange permet de formaliser correctement le fait d'imposer des conditions sur les inconnues primales d'un système linéaire. Elle a toutefois le défaut d'accroître la taille du système à résoudre.

Par ailleurs, la matrice \hat{K} n'est pas toujours factorisable sans pivotage. Dans Cast3M, on utilise la méthode des doubles multiplicateurs de Lagrange pour contourner cette difficulté. Pour plus de détails sur la mise en œuvre de cette méthode, on peut consulter l'article de P. Verpeaux et T. Charras : « Multiplicateur de Lagrange, Condensation Statique et Conditions Unilatérales ».

3.4 Opérateurs de Cast3M associés

Dans Cast3M, ce type de conditions est réalisé grâce aux opérateurs suivants :

- A : BLOQuer et RELAtion ;
- d : DEPI (DEPlacement Imposé) ;
- \hat{K} et \hat{F} : ET, pour assembler les matrices K^e et A et les second membres $(F^S + F^V - B \cdot \sigma^m)$ et d ;
- \hat{U} : RESOudre avec, comme arguments, \hat{K} et \hat{F} .



Chapitre 4

Résolution de l'équilibre

4.1 Méthode

Dans le cas de comportements mécaniques non-linéaires, la résolution de l'équation d'équilibre (3.2) est réalisée de façon itérative, suivant une méthode de point fixe, jusqu'à ce que le résidu soit suffisamment faible selon un critère donné (cf. §1.5).

Les lois de comportement des matériaux étant généralement formulées en incrément ($\dot{\epsilon}^p = \varphi(\dot{\sigma}, p)$, etc.), il est plus pratique de formuler également l'équation d'équilibre en incrément.

Pour cela, considérons des conditions initiales définies par $(U_0, \lambda_0, \sigma_0)$, éventuellement hors équilibre, et cherchons une solution $(U_1, \lambda_1, \sigma_1)$ qui équilibre le *chargement* défini par les efforts et déplacements imposés F_1^S, F_1^V et d_1 . Le passage de l'indice « 0 » à l'indice « 1 » représente un *pas de calcul*.

4.2 Incréments de déplacement

On décompose le champ de déplacement de la façon suivante :

$$U_1^i = U_0 + \Delta U_1^i \quad \text{et} \quad \Delta U_1^{i+1} = \Delta U_1^i + \delta \Delta U_1^{i+1},$$

avec :

- U_0 : le déplacement en début de pas ;
- U_1^i : le déplacement courant ;
- ΔU_1^i : l'incrément courant de déplacement ;
- $\delta \Delta U_1^{i+1}$: la correction de l'incrément courant de déplacement.

On montre facilement que :

$$U_1^{i+1} = U_1^i + \delta \Delta U_1^{i+1}. \quad (4.1)$$

4.3 Formulation incrémentale des conditions sur les déplacements

La formulation de conditions sur les déplacements :

$$A.U_1^{i+1} = d_1,$$

s'écrit aussi de façon incrémentale :

$$A.\delta \Delta U_1^{i+1} = d_1 - A.U_1^i, \quad (4.2)$$

le membre de droite représentant un résidu sur le déplacement imposé.

4.4 Formulation incrémentale de l'équilibre

Après i itérations, on cherche un champ solution $(U_1^{i+1}, \lambda_1^{i+1}, \sigma_1^{i+1})$ qui satisfait l'équation d'équilibre (3.2) :

$$K^e . U_1^{i+1} + A^T . \lambda_1^{i+1} = F_1^S + F_1^V - B . \sigma_1^{nl,i},$$

$\sigma_1^{nl,i}$ étant « la partie non-linéaire » des contraintes σ_1^i obtenues à l'itération précédente. Injectons l'expression incrémentale (4.1) de U_1^{i+1} dans celle-ci :

$$\begin{aligned} K^e . (U_1^i + \delta\Delta U_1^{i+1}) &= F_1^S + F_1^V - A^T . \lambda_1^{i+1} - B . \sigma_1^{nl,i}, \\ K^e . \delta\Delta U_1^{i+1} &= F_1^S + F_1^V - \underbrace{A^T . \lambda_1^{i+1}}_{-F_1^R} - \underbrace{(K^e . U_1^i + B . \sigma_1^{nl,i})}_{B . \sigma_1^i \text{ d'après (2.5)}}. \end{aligned}$$

Au membre de droite, on reconnaît l'expression du résidu, d'où, finalement :

$$K^e . \delta\Delta U_1^{i+1} = R_1^i. \quad (4.3)$$

Dans ces développements, nous n'avons pas précisé la dépendance des opérateurs K^e , B ou A à la solution, ni détaillé ou commenté celle du chargement F_1^S , F_1^V ou F_1^R .

On peut assembler les équations (4.2) et (4.3) pour former le système suivant :

$$\begin{pmatrix} K^e & A^T \\ A & 0 \end{pmatrix} \begin{pmatrix} \delta\Delta U_1^{i+1} \\ \lambda_1^{i+1} \end{pmatrix} = \begin{pmatrix} F_1^S + F_1^V - B . \sigma_1^i \\ d_1 - A . U_1^i \end{pmatrix}. \quad (4.4)$$

En mettant l'équation d'équilibre sous cette forme, la ligne supérieure du second membre n'est plus égale au résidu, qui doit alors être post-traité. La décomposition du multiplicateur de Lagrange en incréments permettrait de préserver l'expression du résidu au second membre mais ce sont les forces de réaction qui devraient alors faire l'objet d'un post-traitement ($F_1^{R,i+1} = F_1^{R,i} - A^T . \delta\lambda^{i+1}$).

4.5 Convergence

Comme évoqué au § 1.5, la résolution itérative de l'équilibre est menée jusqu'à ce que le résidu soit négligeable devant une valeur de référence des efforts mis en jeu dans le problème traité selon une norme donnée :

$$|R| < \zeta F^{ref}. \quad (4.5)$$

Concernant la norme $|\cdot|$, il est classique d'utiliser la norme infinie, à savoir le maximum en valeur absolue de toutes les composantes du résidu. Concernant la valeur de référence F^{ref} , on peut utiliser la même norme appliquée aux efforts extérieurs. Enfin, ζ est la précision, fixée par l'utilisateur.

4.6 Méthode de minimisation du résidu

La résolution itérative de l'équation (4.4) est appelée méthode de « minimisation du résidu ». La figure 4.1 présente un premier algorithme de résolution de l'équilibre basé sur cette méthode. Les notations adoptées ne précisent pas les indices de boucle.

L'étude de la convergence de cet algorithme n'est pas l'objet de ce chapitre. On peut toutefois noter qu'à convergence, la solution $(U_1, \lambda_1, \sigma_1)$ obtenue satisfait l'équilibre des efforts sur le maillage Ω^h à ζ près, pour un chargement F_1^S , F_1^V et d_1 , des paramètres externes p_1 , une raideur K^e et des opérateurs A et B fixés.



Algorithme de minimisation du résidu (1)

$(U_1, \lambda_1) = (U_0, \lambda_0)$
 $(\sigma_1, \varepsilon_1^{in}, \nu_1) = (\sigma_0, \varepsilon_0^{in}, \nu_0)$
 $F_1^R = -A^T \cdot \lambda_1$
 $F^{ref} = |F_1^S + F_1^V + F_1^R|$
tant que $|R_1| \geq \zeta F^{ref}$
 $(\delta\Delta U_1, \lambda_1) = \text{résoudre (4.4)}$
 $U_1 = U_1 + \delta\Delta U_1$
 $\varepsilon_1 = \mathcal{D}(U_1)$
 $(\sigma_1, \varepsilon_1^{in}, \nu_1) = \text{intégrer } C(\varepsilon_1, \varepsilon_1^{in}, \nu_1, p_1)$
 $F_1^R = -A^T \cdot \lambda_1$
 $R_1 = F_1^S + F_1^V + F_1^R - B \cdot \sigma_1$
fin

FIGURE 4.1 – Algorithme de résolution de l'équilibre par minimisation du résidu.

L'opération d'intégration du comportement assure que le résultat $(\sigma_1, \varepsilon_1^{in}, \nu_1)$ satisfait les relations de comportement pour une déformation totale ε_1 mais, généralement, que de manière approchée. Dans ce cas, il est souhaitable que la précision adoptée pour intégrer le comportement soit meilleure que celle utilisée pour vérifier l'équilibre des efforts (ζ) afin d'assurer la stabilité de l'algorithme présenté.

Par ailleurs, il faut également remarquer que, même si l'équilibre des efforts est atteint, rien ne garantit la stabilisation des déformations inélastiques ou des paramètres internes de la loi de comportement d'un itéré sur l'autre de l'algorithme, par exemple que : $|\varepsilon_1^{in,i+1} - \varepsilon_1^{in,i}| < \zeta$. Dans le cas de calculs de fluage, où l'écoulement viscoplastique se produit à effort constant, l'équilibre des efforts peut être atteint sans que le matériau ait terminé de s'écouler. L'ajout d'un critère supplémentaire, vérifiant la stabilisation de la déformation inélastique, est alors indispensable pour garantir la convergence de cet algorithme vers la solution au problème posé.

Enfin, lorsqu'on adopte un point vue Lagrangien, on identifie les points du maillage à ceux du milieu matériel que l'on modélise. Ainsi, la géométrie du maillage évolue au cours du déplacement, de même que les grandeurs intégrées sur ce dernier, notamment F_1^S , F_1^V , K^e , A et B .

L'équilibre statique devant être vérifié dans la configuration du déplacement solution, il conviendrait donc de modifier cet algorithme en mettant à jour les grandeurs associées au maillage au cours des itérations pour garantir sa convergence vers une solution équilibrée dans la configuration déformée. Dans ce cas, on dit qu'on réalise le calcul en *grands déplacements*. Lorsque l'on identifie les différentes configurations du calcul à la configuration de départ, non déformée, on dit qu'on réalise le calcul en *petits déplacements*.

4.7 Opérateurs de Cast3M associés

La méthode de minimisation du résidu est notamment mis en œuvre dans la procédure PASAPAS.



Chapitre 5

Application

5.1 Exemple : statique1.dgibi

5.1.1 Description

On propose de résoudre l'équilibre d'une poutre en flexion simple soumise à son poids propre. Une de ses extrémités est donc encastrée ; l'autre est soumise à un effort tranchant \vec{F} . Sous l'effet de la pesanteur, elle est également soumise à une densité volumique de force $\vec{f} = \rho\vec{g}$. Le problème mécanique ainsi posé est illustré par la figure 5.1.




FIGURE 5.1 – Schéma d'une poutre en flexion simple soumise à son poids propre.

Le comportement mécanique de la poutre est élasto-plastique parfait. Ses caractéristiques mécaniques sont données dans la table 5.1 et ses dimensions, dans la table 5.2. Son épaisseur étant fournie (dimension normale au plan de la figure 5.1), ce problème sera modélisé en trois dimensions.

TABLE 5.1 – Caractéristiques mécaniques de la poutre.

Module d'Young (Gpa)	Coefficient de Poisson	Limite élastique (MPa)	Masse volumique (kg/m ³)
40	0,2	25	2.10 ³

TABLE 5.2 – Dimensions de la poutre.

Longueur (m)	Hauteur (m)	Épaisseur (m)
15	0,6	0,3

5.1.2 Mise en données

La mise en données de ce problème est fournie en annexe A.1, page 23. Ce jeu de données est également disponible sur le Site Cast3M. Pour résoudre l'équilibre, on a programmé l'algorithme de minimisation du résidu présenté au chapitre 4, figure 4.1. La programmation en Gibiane de la boucle de résolution de l'équilibre est présentée à la figure 5.2.

Le problème est résolu en un pas de calcul. Les conditions initiales étant nulles, l'expression du premier résidu se limite à : $R_1 = F_1^V + F_1^S$ (variable R1, ligne 5). On initialise également le champ de déplacement solution par un champ nul (variable U1, ligne 6). La valeur de l'effort de référence F^{ref} est fixée à la première itération de la boucle de résolution (variable Fref1, ligne 20). La précision ζ est définie en tête du jeu de donnée et vaut 1.10^{-4} (variable zprecl, ligne 25).

```

1  *** Resolution iterative par Minimisation du Residu ***
2  * On resout de facon iterative le champ de deplacement solution
3  * en controlant le critere sur l'equilibre des efforts.
4  * En dessous d'une valeur cible (zprecl), on sort :
5  R1      = Fv1 + Fs1 ;
6  U1      = manu chpo V1 3 ux 0. uy 0. uz 0. ;
7  lsml    = extr mol contraintes ;
8  repe b1 20 ;
9  dDul    = reso Kell R1 ;
10 U1      = U1 + dDul ;
11 Fr1     = reac A1 U1 ;
12 eps1    = epsi line mol U1 ;
13 compl   = comp mol mal (eps1 et mal) ;
14 sig1    = exco compl lsml lsml ;
15 bsg1    = bsig mol sig1 ;
16 R1      = Fv1 + Fs1 + Fr1 - bsg1 ;
17 si (&b1 ega 1) ;
18   opti echo 0 ;
19   mess ' Ite.      Critere Forces   Epse' ;
20   Fref1    = maxi abs (Fv1 + Fs1 + Fr1) ;
21 fins ;
22 zcrl    = (maxi abs R1) / Fref1 ;
23 epsel   = exco compl epse epse ;
24 mess &B1 '      ' zcrl '      ' (maxi epsel) ;
25 si (zcrl < zprecl) ;
26   mess ' > Convergence a l iteration ' &B1 ;
27   opti echo 1 ;
28   quit b1 ;
29 fins ;
30 fin b1 ;

```

FIGURE 5.2 – Lignes de Gibiane correspondant à la programmation de la boucle de résolution de l'équilibre par minimisation du résidu (issues du jeu de données fourni en annexe A.1).

5.1.3 Exécution

La figure 5.3 présente les messages affichés par le jeu de données au cours de la résolution de l'équilibre. Ces affichages indiquent que l'équilibre est atteint en 10 itérations. En plus de la valeur du critère sur le résidu (Critère Forces), le jeu de données affiche également la valeur de la déformation plastique cumulée (Epse).



5.1. EXEMPLE : STATIQUE1.DGIBI

```
1  Ite.      Critere Forces      Epse
2      1      3.42647E-02      5.10392E-05
3      2      1.33969E-02      7.02006E-05
4      3      5.41863E-03      7.77741E-05
5      4      2.57575E-03      8.09317E-05
6      5      1.51816E-03      8.24369E-05
7      6      8.08002E-04      8.32120E-05
8      7      4.24111E-04      8.36223E-05
9      8      2.22945E-04      8.38420E-05
10     9      1.17703E-04      8.39601E-05
11    10      6.24183E-05      8.40239E-05
12    > Convergence a 1 iteration      10
```

FIGURE 5.3 – Messages affichés par le jeu de données au cours de la résolution de l'équilibre.

Pour illustrer le résultat du calcul, les figures 5.4 et 5.5 présentent, respectivement, les isovaleurs de la contrainte σ_{xx} et de la déformation plastique cumulée ϵ_{eq}^p obtenues à l'équilibre, affichées sur le maillage déformé avec un facteur d'amplification de 20.




FIGURE 5.4 – Illustration du résultat : contrainte σ_{xx} sur la configuration déformée (x 20).




FIGURE 5.5 – Illustration du résultat : déformation plastique cumulée $\varepsilon_{e_q}^p$ sur la configuration déformée (x 20).



Annexe A

Jeux de données Cast3M

A.1 statique1.dgibi

```
* fichier : statique1.dgibi
*****
* Illustration de la methode de resolution d'un equilibre mecanique *
* par minimisation iterative du residu. *
* *
* On calcule la flexion simple d'une poutre au comportement elasto- *
* plastique parfait en petits déplacements. *
* *
*****

* Pour visualiser les tracer, commentez la ligne ci-dessous :
opti trac psc ;

* Options generales : 3D, hexahedre lineaire
opti dime 3 elem cub8 ;

* Precision :
zprecl = 1.e-4 ;

*** Geometrie et maillage ***
* Dimensions de la poutre (Longuer, Epaisseur, Hauteur) :
l1 = 15. ;
e1 = 0.3 ;
h1 = 0.6 ;

* Repere :
o1 = 0 0 0 ;
x1 = 1 0 0 ;
y1 = 0 1 0 ;
z1 = 0 0 1 ;

* Maillage :
ly1 = o1 droi 4 (e1 * y1) ;
s1 = ly1 tran 8 (h1 * z1) ;
v1 = s1 volu tran 100 (l1 * x1) ;
s2 = v1 face 2 ;
trac qual cach v1
titr (chai 'maillage de la poutre (nb noeuds =' (nbno v1) ')') ;

*** Modele mecanique ***
```

```

* Modele & Caracteristiques materielles :
mol      = mode v1 mecanique elastique plastique parfait ;
mal      = mate mol youn 40.e9 nu 0.2 sigy 25.e6 rho 2.e3 ;

* Conditions sur les déplacements :
A1       = bloq depl s1 ;

* Forces volumiques et surfaciques :
grav1    = manu chml mol fx 0. fy 0. fz -9.81 rigidites type scalaire ;
rho1     = (exco mal rho scal) chan type 'FOR. VOL' ;
Fv1      = cneq mol (rho1 * grav1) ;
Fs1      = forc s2 (0 0 -1.e4) ;
trac     cach (vect Fv1 forc bleu 5.e-2) v1
         titr 'Forces nodales equivalentes volumiques' ;
trac     cach (vect Fs1 forc vert 5.e-3) (aret v1 et s2)
         titr 'Forces nodales equivalentes surfaciques' ;

*** Resolution :
* Rigidites :
Kell     = rigi mol mal ;
Kell     = Kell et A1 ;

*** Resolution iterative par Minimisation du Residu ***
* On resout de facon iterative le champ de deplacement solution
* en controlant le critere sur l'equilibre des efforts.
* En dessous d'une valeur cible (zprecl), on sort :
R1       = Fv1 + Fs1 ;
U1       = manu chpo V1 3 ux 0. uy 0. uz 0. ;
lsml     = extr mol contraintes ;
repe     bl 20 ;
  dDul    = reso Kell R1 ;
  U1      = U1 + dDul ;
  Fr1     = reac A1 U1 ;
  eps1    = epsi line mol U1 ;
  compl   = comp mol mal (eps1 et mal) ;
  sig1    = exco compl lsml lsml ;
  bsg1    = bsig mol sig1 ;
  R1      = Fv1 + Fs1 + Fr1 - bsg1 ;
  si (&bl ega 1) ;
    opti  echo 0 ;
    mess ' Ite.      Critere Forces      Epse' ;
    Fref1  = maxi abs (Fv1 + Fs1 + Fr1) ;
  fins ;
  zcrl    = (maxi abs R1) / Fref1 ;
  epsel   = exco compl epse epse ;
  mess &B1 '      ' zcrl '      ' (maxi epsel) ;
  si (zcrl < zprecl) ;
    mess ' > Convergence a 1 iteration ' &B1 ;
    opti  echo 1 ;
    quit  bl ;
  fins ;
fin bl ;
epsel    = exco compl epse epse ;
*
* Deformees :
defo     = defo U1 (aret v1) 0. blan ;
def1     = defo U1 (enve v1) 20. roug ;

```



A.1. STATIQUE1.DGIBI

```
trac cach (def0 et def1) titr ' Deformee ' ;

* Contrainte SMxx sur deformee :
niso1      = 14 ;
sig1       = 1.e-6 * sig1 ;
defsg1     = defo U1 v1 20. mol (exco sig1 smxx smxx) ;
trac defsg1 titr ' Contrainte SMxx sur deformee (MPa) ' niso1 ;

* Deformations inelastiques EIXX sur deformee :
liso1      = prog 0. pas 0.1e-4 1.2e-4 ;
defe11     = defo U1 v1 20. mol epsel ;
trac defe11 titr ' Deformation plastique cumulee sur deformee ' liso1 ;

fin ;
```

